# Maintainer: Andrzej Giniewicz <gginiu@gmail.com>
# Contributor: Oliver Sherouse <oliver DOT sherouse AT gmail DOT com>
# Contributor: Nick Ward <ward.nickjames@gmail.com>
# Contributor: Rich Lindsley <rich at dranek dot com>

pkgname='python-patsy'
pkgver=0.5.3
pkgrel=2
pkgdesc="A Python package for describing statistical models using symbolic formulas"
arch=('any')
url="https://github.com/pydata/patsy"
license=('BSD')
depends=('python-numpy')
makedepends=('python-setuptools')
optdepends=('python-nose: needed for integrated test suite'
            'python-scipy: needed for spline-related functions')
source=("https://github.com/pydata/patsy/archive/v$pkgver/$pkgname-$pkgver.tar.gz")
sha256sums=('a70452e6b78906240f7e4e04629b5341698476e8a4b4b4fdd63044fd91a70030')

build() {
  cd patsy-${pkgver}
  python setup.py build
}

package() {
  cd patsy-${pkgver}

  python setup.py install --skip-build --root="$pkgdir" --optimize=1

  install -Dm644 LICENSE.txt "$pkgdir"/usr/share/licenses/$pkgname/LICENSE
}
